1

```
(%i1) ext_euclid(a,b):=block(
          [x,y,d,x_old,y_old,d_old],
          if b = 0 then return([1,0,a])
          else ([x_old,y_old,d_old]:ext_euclid(b,mod(a,b)),
                [x,y,d]:[y_old,x_old-quotient(a,b)*y_old,d_old],
                return([x,y,d])))$
```

```
(%i2) ext_euclid(258,192);
(%o2) [3,-4,6]
```

Check:

```
(%i3) 3*258-4*192;
(%o3) 6
```

2

(a) Use Bezout to find the inverse of 3 mod 7, then multiply by 4. Check.

```
(%i4) ext_euclid(3,7)$ %[1];
      %*4$ mod(%,7);
      mod(3*%,7);
(%o5) -2
(%o7) 6
(%o8) 4
```

(b) Chinese remainder formula. 2 checks.

```
(%i9) mods:[3,5,7]$
      rems:[2,3,2]$

      m:lcm(mods);
      mk:%/mods;
      makelist(ext_euclid(%[k],mods[k])[1],k,3);
      sum(%[k]*mk[k]*rems[k],k,1,3)$ mod(%,m);

      chinese(rems,mods);
      mod(%,mods);
(%o11) 105
(%o12) [35,21,15]
(%o13) [-1,1,1]
(%o15) 23
(%o16) 23
(%o17) [2,3,2]
```

3

```
(%i18) makelist([n,sum(2*k-1,k,1,n)],n,9);
(%o18) [[1,1],[2,4],[3,9],[4,16],[5,25],[6,36],[7,49],[8,64],[9,81]]
```

```
Theorem. For all n > 0 the sum of the first n consecutive odd integers is n^2, i.e.
         sum(2k-1,k=1..n)=n^2.

Proof.  Induction on n.

Basis (n=1): 1=1^2. :)

Let n>1.  Assume true for all k<n, in particular, for k=n-1, i.e.
          sum(2k-1,k=1..n-1)=(n-1)^2. Then

          sum(2k-1,k=1..n)=sum(2k-1,k=1..n-1)+2n-1=(n-1)^2+2n-1=n^2-2n+1+2n-1=n^2. :)
```

4

```
(%i19) /* ... (78 lines hidden)
```

```
(%i24) matrix([3,4,0,7],[1,1,1,-7]);
       echelon(%);
       rref(%);
```

(%o24) $\begin{bmatrix} 3 & 4 & 0 & 7 \\ 1 & 1 & 1 & -7 \end{bmatrix}$

(%o25) $\begin{bmatrix} 1 & \frac{4}{3} & 0 & \frac{7}{3} \\ 0 & 1 & -3 & 28 \end{bmatrix}$

(%o26) $\begin{bmatrix} 1 & 0 & 4 & -35 \\ 0 & 1 & -3 & 28 \end{bmatrix}$

```
(%i27) [3*x+4*y=7,x+y+z+7=0];
       solve(%,[x,y]);
```

(%o27) $[4\,y+3\,x=7\,,z+y+x+7=0]$

(%o28) $[[x=-4\,z-35\,,y=3\,z+28]]$

5

```
(%i29) A:matrix([0,1],[0,0]); A.A;
```

(%o29) $\begin{bmatrix} 0 & 1 \\ 0 & 0 \end{bmatrix}$

(%o30) $\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$

```
(%i31) X:matrix([1,1],[0,0]);
charpoly(X,x);
solve(%,x);
eigenvectors(X);
```

(%o31) $\begin{bmatrix} 1 & 1 \\ 0 & 0 \end{bmatrix}$

(%o32) $-(1-x)\,x$

(%o33) $[x=0, x=1]$

(%o34) [[[0,1],[1,1]],[[[1,-1]],[[1,0]]]]

```
(%i35) X:A+ident(2);
charpoly(X,x);
solve(%,x);
eigenvectors(X);
```

(%o35) $\begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix}$

(%o36) $(1-x)^2$

(%o37) $[x=1]$

(%o38) [[[1],[2]],[[[1,0]]]]

```
(%i39) B:matrix([1,0],[0,-1]);
X.B; B.X;
```

(%o39) $\begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$

(%o40) $\begin{bmatrix} 1 & -1 \\ 0 & -1 \end{bmatrix}$

(%o41) $\begin{bmatrix} 1 & 1 \\ 0 & -1 \end{bmatrix}$

6

(%i42)
```
A:matrix([3,5],[1,-1]);
charpoly(%,x);
ev:solve(%,x);
V:A-x*ident(2);
makelist(substitute(ev[k],V),k,2);
makelist(rref(substitute(ev[k],V)),k,2);
eig:eigenvectors(A);
```

(%o42) $\begin{bmatrix} 3 & 5 \\ 1 & -1 \end{bmatrix}$

(%o43) $(-x-1)(3-x)-5$

(%o44) $[x=-2, x=4]$

(%o45) $\begin{bmatrix} 3-x & 5 \\ 1 & -x-1 \end{bmatrix}$

(%o46) $[\begin{bmatrix} 5 & 5 \\ 1 & 1 \end{bmatrix}, \begin{bmatrix} -1 & 5 \\ 1 & -5 \end{bmatrix}]$

(%o47) $[\begin{bmatrix} 1 & 1 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & -5 \\ 0 & 0 \end{bmatrix}]$

(%o48) $[[[-2,4],[1,1]],[[[1,-1]],[[1,\frac{1}{5}]]]]$

(%i49)
```
P:transpose(matrix(eig[2][1][1],eig[2][2][1]));
Pi:invert(P);
P.Pi;
Pi.A.P;
```

(%o49) $\begin{bmatrix} 1 & 1 \\ -1 & \frac{1}{5} \end{bmatrix}$

(%o50) $\begin{bmatrix} \frac{1}{6} & -\frac{5}{6} \\ \frac{5}{6} & \frac{5}{6} \end{bmatrix}$

(%o51) $\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

(%o52) $\begin{bmatrix} -2 & 0 \\ 0 & 4 \end{bmatrix}$

7

Requiring bad and def is equivalent to requiring badef.
Consider badef and 42 as single tokens, i.e. lose 2 and adef as tokens. 11!

```
(%i53) 16-1-4;
       %!;
(%o53) 11
(%o54) 39916800
```

8

Define the sample space and the two events. Compute probabilities.

```
(%i55) S:{LL,LG,GL,GG};
       E:{LG,GL};
       F:{LL,LG,GL};
       pE:cardinality(E)/cardinality(S);
       pF:cardinality(F)/cardinality(S);
       intersection(E,F);
       pEF:cardinality(%)/cardinality(S);
       pE*pF;
```

(%o55) $\{GG, GL, LG, LL\}$

(%o56) $\{GL, LG\}$

(%o57) $\{GL, LG, LL\}$

(%o58) $\frac{1}{2}$

(%o59) $\frac{3}{4}$

(%o60) $\{GL, LG\}$

(%o61) $\frac{1}{2}$

(%o62) $\frac{3}{8}$

Since p(E)*p(F) differs from p(E intersection F), the events are not idependent.

9

Set the total integral to 1 and solve for c. Check.

```
(%i71) f:c*(1-0.025*t); b:40;
       integrate(f,t,0,b);
       solve(%=1,c); float(%);
       f:substitute(%,f);
       integrate(f,t,0,b);
```

(%o71) $c\,(1-0.025\,t)$

(%o72) 40

```
rat: replaced -0.025 by -1/40 = -0.025
rat: replaced 2.0 by 2/1 = 2.0
rat: replaced -0.0125 by -1/80 = -0.0125
rat: replaced 20.0 by 20/1 = 20.0
```

(%o73) $20\,c$

(%o74) $[c=\frac{1}{20}]$

(%o75) $[c=0.05]$

(%o76) $0.05\,(1-0.025\,t)$

```
rat: replaced -0.025 by -1/40 = -0.025
rat: replaced 2.0 by 2/1 = 2.0
rat: replaced -0.0125 by -1/80 = -0.0125
rat: replaced 20.0 by 20/1 = 20.0
```

(%o77) 1.0

```
(%i78) wxplot2d(f,[t,0,b]);
```

(%t78)



(%o78)

```
(%i79) integrate(f,t,0,10);
rat: replaced -0.025 by -1/40 = -0.025
rat: replaced 2.0 by 2/1 = 2.0
rat: replaced -0.0125 by -1/80 = -0.0125
rat: replaced 8.75 by 35/4 = 8.75
(%o79) 0.4375
```

43.75% probability of healing in the first 10 days.

```
(%i80) integrate(t*f,t,0,b);
rat: replaced -0.025 by -1/40 = -0.025
rat: replaced 0.025 by 1/40 = 0.025
rat: replaced -0.025 by -1/40 = -0.025
(%o80) 13.33333333333333
```

On average a tattoo heals in 13.3 days (mean).

```
(%i81) integrate(f,t);
       solve(%=1/2,t);
       float(%);
```

(%o81) $0.05\left(t-0.0125\,t^2\right)$

```
rat: replaced 0.05 by 1/20 = 0.05
rat: replaced -0.0125 by -1/80 = -0.0125
```

(%o82) $[t=40-5\,2^{5/2}, t=5\,2^{5/2}+40]$

(%o83) $[t=11.71572875253808, t=68.28427124746191]$

The second value is out of range, so half the tattoos heal in 11.7 days (median).